
BIYSC 2021 - Notes

Release 0.1

Marc Masdeu
Roberto Rubio

Jul 21, 2021

CONTENTS

1	Introduction	1
2	Logic in Lean - Part 1	3
3	Logic in Lean - Part 2	11
4	Pretty Symbols in Lean	15
5	Glossary of tactics	17

INTRODUCTION

1.1 What is Lean?

Lean is an open source proof-checker and a proof-assistant. One can *explain* mathematical proofs to it and it can check their correctness. It also simplifies the proof writing process by providing *goals* and *tactics*.

Lean is built on top of a formal system called type theory. In type theory, the basic notions are “terms” and “types” — compare to “elements” and “sets” in set theory. Every term has a type, and types are just a special kind of term. Terms can be interpreted as mathematical objects, functions, propositions, or proofs. The only two things Lean can do is *create* terms and *check* their types. By iterating these two operations, we can teach Lean to verify complex mathematical proofs. For instance, below you can see the statement of Fermat’s Last Theorem coded in Lean, which states that the equation $x^n + y^n = z^n$ has no nontrivial solutions in the integers when $n \in \mathbb{N}$ is greater than 2.

```
def x := 2 + 2 -- a natural number
def f (x : ℕ) := x + 3 -- a function
def easy_theorem_statement := 2 + 2 = 4 -- a proposition
def fermats_last_theorem_statement -- another proposition
  :=
  ∀ n : ℕ,
  n > 2
  →
  ¬ (∃ x y z : ℕ, (x^n + y^n = z^n) ∧ (x ≠ 0) ∧ (y ≠ 0) ∧ (z ≠ 0))

theorem
easy_proof : easy_theorem_statement -- proof of easy_theorem
:=
begin
  exact rfl,
end

theorem
my_hard_proof : fermats_last_theorem_statement -- cheating!
:=
begin
  sorry,
end

#check x
#check f
#check easy_theorem_statement
#check fermats_last_theorem_statement
#check easy_proof
#check my_hard_proof
```

1.2 How to use these notes

Every once in a while, you will see a code snippet like this:

```
#eval "Hello, World!"
```

Clicking on the `try it!` button in the upper right corner will open a copy in a window so that you can edit it, and Lean provides feedback in the `Lean Infoview` window. We use this feature to provide exercises inline in the notes. We recommend attempting each exercise as you go along.

These notes are based a 5-day Lean crash course at Mathcamp 2020. We have adapted them to BIYSC 2021.

These notes provide a sneak-peek into the world of theorem proving in Lean and are by no means comprehensive. It is recommended that you simultaneously attempt the [Natural Number Game](#). It is a fun (and highly addictive!) game that proves same basic properties of natural numbers in Lean.

1.3 Acknowledgments.

These notes are based on work of [Apurva Nakade](#) and [Jalex Stark](#). Large chunks of these notes are taken directly from [<https://apurvanakade.github.io/courses/lean_at_MC2020/>](https://apurvanakade.github.io/courses/lean_at_MC2020/)__.

1.4 Useful Links.

1. [Formalizing 100 theorems](#)
2. [Formalizing 100 theorems in Lean](#)
3. **Articles, videos, blog posts, etc.**
 1. [The Xena Project](#)
 2. [The Mechanization of Mathematics](#)
 3. [The Future of Mathematics](#)
4. [Lean Zulip chat group](#)

LOGIC IN LEAN - PART 1

Lean is built on top of a logic system called *type theory*, which is an alternative to *set theory*. In type theory, instead of elements we have *terms* and every term has a *type*. When translated to math, terms can be either mathematical objects, functions, propositions, or proofs. The notation $x : X$ stands for “ x is a term of type X ” or “ x is an inhabitant of X ”. For the most part, you can think of a type as a set and terms as elements of the set.

2.1 Propositions as types

In set theory, a **proposition** is any statement that has the potential of being true or false, like $2 + 2 = 4$, $2 + 2 = 5$, “Fermat’s last theorem”, or “Riemann hypothesis”. In type theory, there is a special type called `Prop` whose inhabitants are propositions. Furthermore, each proposition P is itself a type and the inhabitants of P are its proofs!

```
P : Prop      -- P is a proposition
hp : P        -- hp is a proof of P
```

As such, in type theory “producing a proof of P ” is the same as “producing a term of type P ” and so a proposition P is `true` if there exists a term `hp` of type P .

Notation. Throughout these notes, P , Q , R , \dots will denote propositions.

2.1.1 Implication

In set theory, the proposition $P \Rightarrow Q$ (“ P implies Q ”) is true if either both P and Q are true or if P is false. In type theory, a proof of an implication $P \Rightarrow Q$ is just a function $f : P \rightarrow Q$. Given a function $f : P \rightarrow Q$, every proof `hp` : P produces a proof `f hp` : Q . If P is false then P is *empty*, and there exists an *empty function* from an empty type to any type. Hence, in type theory we use \rightarrow to denote implication.

2.1.2 Negation

In type theory, there is a special proposition `false` : `Prop` which has no proof (hence is *empty*). The negation of a proposition $\neg P$ is the implication $P \rightarrow \text{false}$. Such a function exists if and only if P itself is empty (*empty function*), hence $P \rightarrow \text{false}$ is inhabited if and only if P is empty which justifies using it as the definition of $\neg P$.

To summarize:

1. Proving a proposition P is equivalent to producing an inhabitant `hp` : P .
2. Proving an implication $P \rightarrow Q$ is equivalent to producing a function `f` : $P \rightarrow Q$.
3. The negation, $\neg P$, is defined as the implication $P \rightarrow \text{false}$.

2.1.3 Propositions in Lean

In Lean, a proposition and its proof are written using the following syntax.

```
theorem fermats_last_theorem
  (n : ℕ)
  (n_gt_2 : n > 2)
  :
  ¬ (∃ x y z : ℕ, (x^n + y^n = z^n) ∧ (x ≠ 0) ∧ (y ≠ 0) ∧ (z ≠ 0))
:=
begin
  sorry,
end
```

Let us parse the above statement.

- `fermats_last_theorem` is the name of the theorem.
- `(n : ℕ)` and `(n_gt_2 : n > 2)` are the two *hypotheses*. The former says `n` is a natural number and the latter says that `n_gt_2` is a proof of `n > 2`.
- `:` is the delimiter between hypotheses and targets
- `¬ (∃ x y z : ℕ, (x^n + y^n = z^n) ∧ (x ≠ 0) ∧ (y ≠ 0) ∧ (z ≠ 0))` is the *target* of the theorem.
- `:= begin ... end` contains the proof. When you start your proof, Lean opens up a goal window for you to keep track of hypotheses and targets. **Your goal is to produce a term that has the type of the target.**

```
-- example of Lean goal window
n : ℕ, -- hypothesis 1
n_gt_2 : n > 2 -- hypothesis 2
⊢ ¬ ∃ (x y z : ℕ), x ^ n + y ^ n = z ^ n ∧ x ≠ 0 ∧ y ≠ 0 ∧ z ≠ 0 -- target
```

- The commands you write between `begin` and `end` are called *tactics*. `sorry`, is an example of a tactic. **Very Important:** All tactics must end with a comma (`,`).

Even though they are not explicitly displayed, all the theorems in the Lean library are also hypotheses that you can use to close the goal.

2.2 Implications in Lean

We'll start learning tactics by proving implications in Lean. In the following sections, there are tables describing what a tactic does. Solve the following exercises to see the tactics in action.

The first two tactics we'll learn are `exact` and `intros`.

<code>exact</code>	If <code>P</code> is the target of the current goal and <code>hp</code> is a term of type <code>P</code> , then <code>exact hp</code> , will close the goal. Mathematically, this saying “this is <i>exactly</i> what we were required to prove”.
<code>intro</code>	If the target of the current goal is a function <code>P → Q</code> , then <code>intro hp</code> , will produce a hypothesis <code>hp : P</code> and change the target to <code>Q</code> . Mathematically, this is saying that in order to define a function from <code>P</code> to <code>Q</code> , we first need to choose an arbitrary element of <code>P</code> .


```

/-----

``exact``

If ``P`` is the target of the current goal and
``hp`` is a term of type ``P``, then
``exact hp,`` will close the goal.

``intro``

If the target of the current goal is a function ``P → Q``, then
``intro hp,`` will produce a hypothesis
``hp : P`` and change the target to ``Q``.

Delete the ``sorry,`` below and replace them with a legitimate proof.

-----/

theorem tautology (P : Prop) (hp : P) : P :=
begin
  sorry,
end

theorem tautology' (P : Prop) : P → P :=
begin
  sorry,
end

example (P Q : Prop) : (P → (Q → P)) :=
begin
  sorry,
end

-- Can you find two different ways of proving the following?
example (P Q : Prop) : ((Q → P) → (Q → P)) :=
begin
  sorry,
end

```

The next two tactics are `have` and `apply`.

have	<p><code>have</code> is used to create intermediate variables.</p> <p>If f is a term of type $P \rightarrow Q$ and hp is a term of type P, then <code>have hq := f (hp),</code> creates the hypothesis $hq : Q$.</p>
apply	<p><code>apply</code> is used for backward reasoning.</p> <p>If the target of the current goal is Q and f is a term of type $P \rightarrow Q$, then <code>apply f,</code> changes target to P.</p> <p>Mathematically, this is equivalent to saying “because P implies Q, to prove Q it suffices to prove P”.</p>

Often these two tactics can be used interchangeably. Think of `have` as reasoning forward and `apply` as reasoning backward. When writing a big proof, you often want a healthy combination of the two that makes the proof readable.

```

/-----

```

(continues on next page)

(continued from previous page)

```

``have``

If ``f`` is a term of type ``P → Q`` and
``hp`` is a term of type ``P``, then
``have hq := f(hp),`` creates the hypothesis ``hq : Q`` .

``apply``

If the target of the current goal is ``Q`` and
``f`` is a term of type ``P → Q``, then
``apply f,`` changes target to ``P``.

Delete the ``sorry,`` below and replace them with a legitimate proof.

-----/

example (P Q R : Prop) (hp : P) (f : P → Q) (g : Q → R) : R :=
begin
  sorry,
end

example (P Q R S T U : Type)
(hpq : P → Q)
(hqr : Q → R)
(hqt : Q → T)
(hst : S → T)
(htu : T → U)
: P → U :=
begin
  sorry,
end

```

For the following exercises, recall that $\neg P$ is defined as $P \rightarrow \text{false}$, $\neg (\neg P)$ is $(P \rightarrow \text{false}) \rightarrow \text{false}$, and so on.

```

/-----

Recall that
  ``¬ P`` is ``P → false``,
  ``¬ (¬ P)`` is ``(P → false) → false``, and so on.

Delete the ``sorry,`` below and replace them with a legitimate proof.

-----/

theorem self_imp_not_not_self (P : Prop) : P → ¬ (¬ P) :=
begin
  sorry,
end

theorem contrapositive (P Q : Prop) : (P → Q) → (¬Q → ¬P) :=
begin
  sorry,
end

```

(continues on next page)

(continued from previous page)

```
example (P : Prop) : ¬ (¬ (¬ P)) → ¬ P :=
begin
  sorry,
end
```

2.3 Proof by contradiction

You can prove exactly one of the converses of the above three using just `exact`, `intro`, `have`, and `apply`. Can you find which one?

```

/-----

You can prove exactly one of the following three using just
`exact`, `intro`, `have`, and `apply`.

Can you find which one?

-----/

theorem not_not_self_imp_self (P : Prop) : ¬ ¬ P → P :=
begin
  sorry,
end

theorem contrapositive_converse (P Q : Prop) : (¬Q → ¬P) → (P → Q) :=
begin
  sorry,
end

example (P : Prop) : ¬ P → ¬ ¬ ¬ P :=
begin
  sorry,
end
```

This is because it is not true that $\neg \neg P = P$ by *definition*, after all, $\neg \neg P$ is $(P \rightarrow \text{false}) \rightarrow \text{false}$ which is drastically different from P . There is an extra axiom called **the law of excluded middle** which says that either P is inhabited or $\neg P$ is inhabited (and there is no *middle* option) and so $P \leftrightarrow \neg \neg P$. This is the axiom that allows for proofs by contradiction. Lean provides us the following tactics to use it.

ex-falso	Changes the target of the current goal to false. The name derives from “ <i>ex falso, quodlibet</i> ” which translates to “from contradiction, anything”. You should use this tactic when there are contradictory hypotheses present.
by_cases	If $P : \text{Prop}$, then <code>by_cases P</code> , creates two goals, the first with a hypothesis $hp : P$ and second with a hypothesis $hp : \neg P$. Mathematically, this is saying either P is true or P is false. <code>by_cases</code> is the most direct application of the law of excluded middle.
by_contradiction	If the target of the current goal is Q , then <code>by_contradiction</code> , changes the target to false and adds $hnq : \neg Q$ as a hypothesis. Mathematically, this is proof by contradiction.
push_neg	<code>push_neg</code> , simplifies negations in the target. For example, if the target of the current goal is $\neg \neg P$, then <code>push_neg</code> , simplifies it to P . You can also push negations across a hypothesis $hp : P$ using <code>push_neg at hp</code> .
contrapose!	If the target of the current goal is $P \rightarrow Q$, then <code>contrapose!</code> , changes the target to $\neg Q \rightarrow \neg P$. If the target of the current goal is Q and one of the hypotheses is $hp : P$, then <code>contrapose! hp</code> , changes the target to $\neg P$ and changes the hypothesis to $hp : \neg Q$. Mathematically, this is replacing the target by its contrapositive.

Even though the list is long, these tactics are almost all *obvious*. The only two slightly unusual tactics are `exfalso` and `by_cases`. You’ll see `by_cases` in action later. For the following exercises, you only require `exfalso`, `push_neg`, and `contrapose!`.

```

/-----

``exfalso``

  Changes the target of the current goal to ``false``.

``push_neg``

  ``push_neg`` simplifies negations in the target.
  You can push negations across a hypothesis ``hp : P`` using
  ``push_neg at hp``.

``contrapose!``

  If the target of the current goal is ``P → Q``,
  then ``contrapose!`` changes the target to ``¬ Q → ¬ P``.

  If the target of the current goal is ``Q`` and
  one of the hypotheses is ``hp : P``, then
  ``contrapose! hp`` changes the target to ``¬ P`` and
  changes the hypothesis to ``hp : ¬ Q``.

Delete the ``sorry`` below and replace them with a legitimate proof.

-----/

theorem not_not_self_imp_self (P : Prop) : ¬ ¬ P → P :=
begin
  sorry,
end

```

(continues on next page)

(continued from previous page)

```

theorem contrapositive_converse (P Q : Prop) : (¬Q → ¬P) → (P → Q) :=
begin
  sorry,
end

example (P : Prop) : ¬ P → ¬ ¬ ¬ P :=
begin
  sorry,
end

theorem principle_of_explosion (P Q : Prop) : P → (¬ P → Q) :=
begin
  sorry,
end

```

2.4 Geometry

Finally, let's do some geometry! We will introduce the incidence axioms, and start proving some lemmas from them.

```

constants Point Line : Type*
constant belongs : Point → Line → Prop
local notation A `∈` L := belongs A L
local notation A `∉` L := ¬ belongs A L

```

Here is how we can introduce axioms.

```

-- I1: there is a unique line passing through two distinct points.
axiom I1 (A B : Point) (h : A ≠ B) : ∃! (ℓ : Line) , A ∈ ℓ ∧ B ∈ ℓ

-- I2: any line contains at least two points.
axiom I2 (ℓ : Line) : ∃ A B : Point, A ≠ B ∧ A ∈ ℓ ∧ B ∈ ℓ

-- I3: there exists 3 non-collinear points.
axiom I3 : ∃ A B C : Point, (A ≠ B ∧ A ≠ C ∧ B ≠ C ∧ (∀ ℓ : Line, (A ∈ ℓ ∧ B ∈ ℓ) →
  ¬ (C ∈ ℓ) )))

```

Axiom I3 really says that there are 3 non-collinear points. We can actually define what it means to be collinear and prove a statement which is easier to remember.

```

-- We can make our own definitions
def collinear (A B C : Point) : Prop := ∃ (ℓ : Line), (A ∈ ℓ ∧ B ∈ ℓ ∧ C ∈ ℓ)

-- So let's prove that axiom I3 really says that there are 3 non-collinear points
example : ∃ A B C : Point, ¬ collinear A B C :=
begin
  sorry
end

```

In the morning we proved quite in detail the following theorem (we called Theorem 1). Before trying to prove it, make sure that the *Lean* statement is really what the English sentence says.

```

-- Two distinct lines meet at most at one point
example (r s : Line) (h : r ≠ s) (A B : Point) : A ∈ r ∧ B ∈ r ∧ A ∈ s ∧ B ∈ s → A = B :=

```

(continues on next page)

(continued from previous page)

```
begin
  sorry
end
```

Let's prove another useful lemma: given a line, there is a point outside it.

```
-- Use I3 to prove the following lemma
lemma exists_point_not_on_line (ℓ : Line): ∃ A : Point, A ∉ ℓ :=
begin
  sorry
end

-- Challenge: is it true for two lines? If so, prove it
lemma exists_point_not_on_two_line (r s : Line): ∃ A : Point, A ∉ r ∧ A ∉ s :=
begin
  sorry
end
```

LOGIC IN LEAN - PART 2

Your mission today is to wrap up the remaining bits of logic and move on to doing some “actual math”. Remember to **always save your work**. You might find the [Glossary of tactics](#) page and the [Pretty symbols](#) page useful.

Before we move on to new stuff, let’s understand what we did yesterday.

3.1 Behind the scenes

A note on brackets: It is not uncommon to compose half a dozen functions in Lean. The brackets get really messy and unwieldy. As such, Lean will often drop the brackets by following the following conventions.

- The function $P \rightarrow Q \rightarrow R \rightarrow S$ stands for $P \rightarrow (Q \rightarrow (R \rightarrow S))$.
- The expression $a + b + c + d$ stands for $((a + b) + c) + d$.

An easy way to remember this is that, arrows are bracketed on the right and binary operators on the left.

3.1.1 Proof irrelevance

It might feel a bit weird to say that a proposition has proofs as its inhabitants. Proofs can get huge and it seems unnecessary to have to remember not just the statement but also its proof. This is something we don’t normally do in math. To hide this complication, in type theory there is an axiom, called *proof irrelevance*, which says that if $P : \text{Prop}$ and $hp1\ hp2 : P$ then $hp1 = hp2$. Taking our *analogy* with sets further, you can think of a proposition as a set which is either empty or contains a single element (false or true). In fact, in some forms of type theory (e.g. [homotopy type theory](#)) this is taken as the definition of propositions. This is of course not true for general types. For example, $0 : \mathbb{N} \neq 1 : \mathbb{N}$.

3.1.2 Proofs as functions

Every time you successfully construct a proof of a theorem say

```
theorem tautology (P : Prop) : P → P :=
begin
  intro hp,
  exact hp,
end
```

Lean constructs a *proof term* `tautology : $\forall P : \text{Prop}, P \rightarrow P$` (you can see this by typing `#check tautology`).

In type theory, the *for all* quantifier, \forall , is a generalized function, called a [dependent function](#). For all practical purposes, we can think of `tautology` as having the type $(P : \text{Prop}) \rightarrow (P \rightarrow P)$. Note that this is not a function in

the classical sense of the word because the codomain $(P \rightarrow P)$ *depends* on the input variable P . If $Q : \text{Prop}$, then $\text{tautology } (Q)$ is a term of type $Q \rightarrow Q$.

Consider a theorem with multiple hypothesis, say

```
theorem hello_world (hp : P) (hq : Q) (hr : R) : S
```

Once we provide a proof of it, Lean will create a proof term `hello_world : (hp:P) → (hq:Q) → (hr:R) → S`. So that if we have terms `hp' : P, hq' : Q, hr' : R` then `hello_world hp' hq' hr'` (note the convenient lack of brackets) will be a term of type S .

Once constructed, any term can be used in a later proof. For example,

```
example (P Q : Prop) : (P → Q) → (P → Q) :=
begin
  exact tautology (P → Q),
end
```

This is how Lean simulates mathematics. Every time you prove a theorem using tactics a *proof term* gets created. Because of proof irrelevance, Lean forgets the exact content of the proof and only remembers its type. All the proof terms can then be used in later proofs. All of this falls under the giant umbrella of the [Curry–Howard correspondence](#).

We'll now continue our study of the remaining logical operators: *and* (\wedge), *or* (\vee), *if and only if* (\leftrightarrow), *for all* (\forall), *there exists* (\exists).

3.2 And / Or

The operators *and* (\wedge) and *or* (\vee) are very easy to use in Lean. Given a term `hpq : P ∧ Q`, there are tactics that let you create terms `hp : P` and `hq : Q`, and vice versa. Similarly for $P \vee Q$, with a subtle change (see below).

Note that when multiple goals are open, you are trying to solve the topmost goal.

cases	<p><code>cases</code> is a general tactic that breaks a complicated term into simpler ones.</p> <p>If <code>hpq</code> is a term of type $P \wedge Q$, then <code>cases hpq</code> with <code>hp hq</code>, breaks it into <code>hp : P</code> and <code>hq : Q</code>.</p> <p>If <code>fg</code> is a term of type $P \leftrightarrow Q$, then <code>cases fg</code> with <code>f g</code>, breaks it into <code>f : P → Q</code> and <code>g : Q → P</code>.</p> <p>If <code>hpq</code> is a term of type $P \vee Q$, then <code>cases hpq</code> with <code>hp hq</code>, creates two goals and adds the hypotheses <code>hp : P</code> and <code>hq : Q</code> to one each.</p>
split	<p><code>split</code> is a general tactic that breaks a complicated goal into simpler ones.</p> <p>If the target of the current goal is $P \wedge Q$, then <code>split</code>, breaks up the goal into two goals with targets P and Q.</p> <p>If the target of the current goal is $P \times Q$, then <code>split</code>, breaks up the goal into two goals with targets P and Q.</p> <p>If the target of the current goal is $P \leftrightarrow Q$, then <code>split</code>, breaks up the goal into two goals with targets $P \rightarrow Q$ and $Q \rightarrow P$.</p>
left	If the target of the current goal is $P \vee Q$, then <code>left</code> , changes the target to P .
right	If the target of the current goal is $P \vee Q$, then <code>right</code> , changes the target to Q .

```
/-----
`cases`

`cases` is a general tactic that breaks up complicated terms.
```

(continues on next page)

(continued from previous page)

If `hpq` is a term of type `P ∧ Q` or `P ∨ Q` or `P ↔ Q`, then use `cases hpq with hp hq`.

`split`

If the target of the current goal is `P ∧ Q` or `P ↔ Q`, then use `split`.

`left`/`right`

If the target of the current goal is `P ∨ Q`, then use either `left` or `right` (choose wisely).

`exfalso`

Changes the target of the current goal to `false`.

Delete the `sorry` below and replace them with a legitimate proof.

-----/

```

example (P Q : Prop) : P ∧ Q → Q ∧ P :=
begin
  sorry,
end

example (P Q : Prop) : P ∨ Q → Q ∨ P :=
begin
  sorry,
end

example (P Q R : Prop) : P ∧ false ↔ false :=
begin
  sorry,
end

theorem principle_of_explosion (P Q : Prop) : P ∧ ¬ P → Q :=
begin
  sorry,
end

```

3.3 Quantifiers

As mentioned in the introduction the *for all* quantifier, \forall , is a generalization of a function. As such the tactics for dealing with \forall are the same as those for \rightarrow .

have	If <code>hp</code> is a term of type $\forall x : X, P x$ and <code>y</code> is a term of type <code>X</code> then <code>have hpy := hp (y)</code> creates a hypothesis <code>hpy : P y</code> .
intro	If the target of the current goal is $\forall x : X, P x$, then <code>intro x</code> , creates a hypothesis <code>x : X</code> and changes the target to <code>P x</code> .

The *there exists* quantifier, \exists , in type theory is very intuitive. If you want to prove a statement $\exists x : X, P x$ then you need to provide a witness. If you have a term `hp : ∃ x : X, P x` then from this you can extract a witness.

cases	If hp is a term of type $\exists x : X, P x$, then <code>cases hp</code> with x key, breaks it into $x : X$ and $key : P x$.
use	If the target of the current goal is $\exists x : X, P x$ and y is a term of type X , then <code>use y</code> , changes the target to $P y$ and tries to close the goal.

3.4 Geometry

Now it's your turn! Introduce Hilbert's axioms for between-ness. We'll give you the ones for incidence from yesterday.

```
import tactic
constants Point Line : Type*
constant belongs : Point → Line → Prop
local notation A `∈` L := belongs A L
local notation A `∉` L := ¬ belongs A L

-- I1: there is a unique line passing through two distinct points.
axiom I1 (A B : Point) (h : A ≠ B) : ∃! (ℓ : Line), A ∈ ℓ ∧ B ∈ ℓ

-- I2: any line contains at least two points.
axiom I2 (ℓ : Line) : ∃ A B : Point, A ≠ B ∧ A ∈ ℓ ∧ B ∈ ℓ

-- I3: there exists 3 non-collinear points.
axiom I3 : ∃ A B C : Point, (A ≠ B ∧ A ≠ C ∧ B ≠ C ∧ (∀ ℓ : Line, (A ∈ ℓ ∧ B ∈ ℓ) → ¬
  → (C ∈ ℓ)))

-- We can make our own definitions
def collinear (A B C : Point) : Prop := ∃ (ℓ : Line), (A ∈ ℓ ∧ B ∈ ℓ ∧ C ∈ ℓ)

-- We define the between-ness relation
constant between : Point → Point → Point → Prop
local notation A `*` B `*` C := between A B C
```

PRETTY SYMBOLS IN LEAN

To produce a pretty symbol in Lean, type the *editor shortcut* followed by space or tab.

Unicode	Editor Shortcut	Definition
\rightarrow	<code>\to</code>	function or implies
\leftrightarrow	<code>\iff</code>	if and only if
\leftarrow	<code>\l</code>	used by the <code>rw</code> tactic
\neg	<code>\not</code>	negation operator
\wedge	<code>\and</code>	and operator
\vee	<code>\or</code>	or operator
\exists	<code>\exists</code>	there exists quantifier
\forall	<code>\forall</code>	for all quantifier
\mathbb{N}	<code>\nat</code>	type of natural numbers
\mathbb{Z}	<code>\int</code>	type of integers
\circ	<code>\circ</code>	composition of functions
\neq	<code>\ne</code>	not equal to
\in	<code>\in</code>	belongs to
\notin	<code>\notin</code>	does not belong to
\angle	<code>\angle</code>	angle
\triangle	<code>\triangle</code>	triangle
\cong	<code>\cong</code>	congruence of segments
\simeq	<code>\simeq</code>	congruence of angles

GLOSSARY OF TACTICS

5.1 Implications in Lean

<code>exact</code>	<p>If P is the target of the current goal and hp is a term of type P, then <code>exact hp</code>, will close the goal.</p> <p>Mathematically, this saying “this is <i>exactly</i> what we were required to prove”.</p>
<code>intro</code>	<p>If the target of the current goal is a function $P \rightarrow Q$, then <code>intro hp</code>, will produce a hypothesis $hp : P$ and change the target to Q.</p> <p>Mathematically, this is saying that in order to define a function from P to Q, we first need to choose an arbitrary element of P.</p>

<code>have</code>	<p><code>have</code> is used to create intermediate variables.</p> <p>If f is a term of type $P \rightarrow Q$ and hp is a term of type P, then <code>have hq := f hp</code>, creates the hypothesis $hq : Q$.</p>
<code>apply</code>	<p><code>apply</code> is used for backward reasoning.</p> <p>If the target of the current goal is Q and f is a term of type $P \rightarrow Q$, then <code>apply f</code>, changes target to P.</p> <p>Mathematically, this is equivalent to saying “because P implies Q, to prove Q it suffices to prove P”.</p>

5.2 Proof by contradiction

ex-falso	Changes the target of the current goal to false. The name derives from “ <i>ex falso, quodlibet</i> ” which translates to “from contradiction, anything”. You should use this tactic when there are contradictory hypotheses present.
by_cases	If $P : \text{Prop}$, then <code>by_cases P</code> , creates two goals, the first with a hypothesis $hp : P$ and second with a hypothesis $hp : \neg P$. Mathematically, this is saying either P is true or P is false. <code>by_cases</code> is the most direct application of the law of excluded middle.
by_contradiction	If the target of the current goal is Q , then <code>by_contradiction</code> , changes the target to false and adds $hnq : \neg Q$ as a hypothesis. Mathematically, this is proof by contradiction.
push_neg	<code>push_neg</code> , simplifies negations in the target. For example, if the target of the current goal is $\neg \neg P$, then <code>push_neg</code> , simplifies it to P . You can also push negations across a hypothesis $hp : P$ using <code>push_neg at hp</code> .
contrapose!	If the target of the current goal is $P \rightarrow Q$, then <code>contrapose!</code> , changes the target to $\neg Q \rightarrow \neg P$. If the target of the current goal is Q and one of the hypotheses is $hp : P$, then <code>contrapose! hp</code> , changes the target to $\neg P$ and changes the hypothesis to $hp : \neg Q$. Mathematically, this is replacing the target by its contrapositive.

5.3 And / Or

cases	<code>cases</code> is a general tactic that breaks a complicated term into simpler ones. If hpq is a term of type $P \wedge Q$, then <code>cases hpq with hp hq</code> , breaks it into $hp : P$ and $hq : Q$. If hpq is a term of type $P \times Q$, then <code>cases hpq with hp hq</code> , breaks it into $hp : P$ and $hq : Q$. If fg is a term of type $P \leftrightarrow Q$, then <code>cases fg with f g</code> , breaks it into $f : P \rightarrow Q$ and $g : Q \rightarrow P$. If hpq is a term of type $P \vee Q$, then <code>cases hpq with hp hq</code> , creates two goals and adds the hypotheses $hp : P$ and $hq : Q$ to one each.
split	<code>split</code> is a general tactic that breaks a complicated goal into simpler ones. If the target of the current goal is $P \wedge Q$, then <code>split</code> , breaks up the goal into two goals with targets P and Q . If the target of the current goal is $P \times Q$, then <code>split</code> , breaks up the goal into two goals with targets P and Q . If the target of the current goal is $P \leftrightarrow Q$, then <code>split</code> , breaks up the goal into two goals with targets $P \rightarrow Q$ and $Q \rightarrow P$.
left	If the target of the current goal is $P \vee Q$, then <code>left</code> , changes the target to P .
right	If the target of the current goal is $P \vee Q$, then <code>right</code> , changes the target to Q .

5.4 Quantifiers

have	If hp is a term of type $\forall x : X, P\ x$ and y is a term of type Y then <code>have hpy := hp (y)</code> creates a hypothesis $hpy : P\ y$.
intro	If the target of the current goal is $\forall x : X, P\ x$, then <code>intro x</code> , creates a hypothesis $x : X$ and changes the target to $P\ x$.

cases	If hp is a term of type $\exists x : X, P\ x$, then <code>cases hp with x key</code> , breaks it into $x : X$ and $key : P\ x$.
use	If the target of the current goal is $\exists x : X, P\ x$ and y is a term of type X , then <code>use y</code> , changes the target to $P\ y$ and tries to close the goal.

5.5 Proving “trivial” statements

norm_num	<code>norm_num</code> is Lean’s calculator. If the target has a proof that involves <i>only</i> numbers and arithmetic operations, then <code>norm_num</code> will close this goal. If $hp : P$ is an assumption then <code>norm_num at hp</code> , tries to use <code>simplify hp</code> using basic arithmetic operations.
ring	<code>ring</code> , is Lean’s symbolic manipulator. If the target has a proof that involves <i>only</i> algebraic operations, then <code>ring</code> , will close the goal. If $hp : P$ is an assumption then <code>ring at hp</code> , tries to use <code>simplify hp</code> using basic algebraic operations.
linarith	<code>linarith</code> , is Lean’s inequality solver.
simp	<code>simp</code> , is a very complex tactic that tries to use theorems from the <code>mathlib</code> library to close the goal. You should only ever use <code>simp</code> , to <i>close a goal</i> because its behavior changes as more theorems get added to the library.

5.6 Equality

refl	If the current goal is of the form $X = X$ or $P \leftrightarrow P$, then <code>refl</code> will finish the proof. As long as both sides are defined to be equal, this will work. For example, it will work with the goal $3 = 2 + 1$ because <i>by definition</i> the number 3 is defined to be 2 plus one. Mathematically, this says “check that both sides are equal <i>by definition</i> ”.
rw	If f is a term of type $P = Q$ (or $P \leftrightarrow Q$), then <code>rw f</code> , searches for P in the target and replaces it with Q . <code>rw <f</code> , searches for Q in the target and replaces it with P . If additionally, $hr : R$ is a hypothesis, then <code>rw f at hr</code> , searches for P in the expression R and replaces it with Q . <code>rw <f at hr</code> , searches for Q in the expression R and replaces it with P . Mathematically, this is saying because $P = Q$, we can replace P with Q (or the other way around).